
News  
Meetings and events 

8th Meeting of the French Neuroscience Society 2007                                                                
Corum-Esplanade Charles de Gaulle 
Montpellier, France 
May 22nd – 25th 2007 

Life Sciences 2007                                                                                                                    
Scotland Exhibition and Convention Centre 
Glasgow  
July 8th – 12th 2007 

IBRO World Congress of Neuroscience                                                                                                           
Melbourne Exhibition and Convention Centre 
Melbourne, Australia  
July 12th – 16th 2007 

IBRO Satellite Meeting: Motor Control at the Top End                                                                                                            
Darwin, Australia  
July 18th – 21st 2007 

 

  Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.  
 

 Spike2 downloads Signal downloads 
Spike2 version 6.03  Signal version 3.09 

 Spike2 version 5.17  Signal version 2.16 

 Spike2 demo  Signal demo 

A full list of the new features and changes in the software versions is available from 
the website. 

Spike2   

 

Q. I use Spike2 for acquisition and analysis on a PC but would also ideally like to 
analyse files on my office computer, which is a new Intel-based Mac. I have 
looked at using Apple’s Boot Camp to install Windows on my Mac so that I can 
run Spike2. Do you have any recommendations at all? 

A. Apple’s Boot Camp technology allows owners of Intel-based Mac systems to 
install and run a copy of Windows XP or Vista in addition to an existing Mac OS 
X installation. This creates a dual-boot system from which the user can select  

 between running OS X or Windows. We have used Boot Camp to install 
Windows XP on a Mac mini running OS X with an Intel Core-Duo 1.66GHz 
processor and 1GB of RAM. Spike2 was then installed with the standard 1401 
drivers and data was sampled successfully. 

The next major release of OS X will include Boot Camp as standard; users of the 
current version of OS X can download Boot Camp 1.2 beta and an installation 
guide from the Apple website. 

     

  

 

Q. I use multiple states in dynamic outputs mode to set-up a series of stimulus 
protocols with varying pulse lengths, intervals and amplitudes. It can be tedious 
to set each individual state from scratch so I was wondering if there was any way 
to copy pulse information between states? 

A. The pulses configuration dialog now includes a Copy button that can be used to 
duplicate outputs between states when using dynamic outputs mode. This makes 
it easier to set up a number of similar pulse output protocols. You select which 
output ‘tracks’ to use and copy the selected information to a range of states. 

  

 

 
Pulse configuration dialog with Copy output options 

Scripts: Spike2  
Q. I regularly use memory channels to mark the times of peaks in my data but often 

find that my initial settings for the amplitude field do not capture all peaks in the 
file if the amplitude changes over time. This means I have to clear data from the 
memory channel and re-process, often more than once. It would be helpful if 
there was some way of interactively adjusting the amplitude level and updating 
the display of the imported events before confirming that the amplitude level was 
correct. 

A. The attached script, FeatureDetect.s2s, allows the user to interactively 
import peaks, troughs or threshold crossings based on an amplitude value 
defined with a pair of horizontal cursors or a single horizontal cursor used as a 
threshold crossing. The script contains an idle routine that automatically updates 
the memory channel holding the event markers if the user adjusts the amplitude 
or threshold level at any time.  

'$FeatureDetect|Interactive feature detect script. Updates imported events based
'on threshold or amplitude settings defined by Horizontal cursor positions.

var data%;      'Data file
var wChan%;     'Waveform channel
var pChan%;     'Memory channel for imported events
var sTime:= 0;  'Start time for import
var eTime;      'End time for import
var n%;             
var o1Lev,o2Lev;    'HCursor levels
var num$;       'Count of events    
var mode%;      'Import mode
var minInt;     'Min Interval

data% := FrontView();       'Bring data file to the front
if ViewKind(data%)<>0 then  'If no data file
    Message("Open a data file for analysis");   'prompt user
        data%:= FileOpen("",0,1);
    if data% < 0 then       'If unable to open file
        Message("Unable to open a data file!");     'Warn user and quit script
        Halt;endif;
endif;

Window(0,0,100,100);        'Make the file full screen

ToolbarSet(1, "Import Features",GetFeats%);     'Set up toolbar
ToolbarSet(3, "Done", Done%);
ToolbarSet(5, "Change Min interval",ChangeInt%);    
Toolbar("Interactive feature detection",1023);


Func GetFeats%();
var ok%;
DlgCreate("Channel selection");
DlgChan(1,"Waveform channel for feature extraction",513);
DlgList(2,"Import method","Peak|Trough|Rising threshold|Falling threshold");
DlgReal(3,"Minimum interval",0,10); 
ok% := DlgShow(wChan%,mode%,minInt);        'Set channel to use and method of detection
if ok% = 0 then
    Halt;       'Halt script if user cancels
endif;

eTime := MaxTime();     'Set end time as end of file
pChan% := MemChan(3);   'Create memory channel to hold events
ChanShow(pChan%);       'Show the memory channel
HCursorDelete(-1);      'Clear any existing horizontal cursors

if mode% < 2 then       'If import mode requires an amplitude setting
    HCursorNew(wChan%,YLow(wChan%) + ((YHigh(wChan%)-YLow(wChan%))*.67));   'Set-up horizontal cursors evenly spaced
    HCursorNew(wChan%,YLow(wChan%) + ((YHigh(wChan%)-YLow(wChan%))*.33));
    n% := Import%(HCursor(1)-HCursor(2));
else
    HCursorNew(wChan%,(YHigh(wChan%)-YLow(wChan%))*0.5);    'else use a single cursor as a threshold
    n% := Import%(HCursor(1));
endif;
num$ := Print$("%d Features imported",n%);
ToolbarText(num$);
o1Lev := HCursor(1);        'Get positions of cursors
o2Lev := HCursor(2);
ToolbarSet(0,"Idle", Idle%);    'Start Idle routine
return 1;
end;


Func Idle%()
View(data%);
if mode%< 2 then        'If import mode requires an amplitude
    if HCursor(1) <> o1Lev or HCursor(2) <> o2Lev then  'Check if positions have changed
    n% := Import%(HCursor(1)-HCursor(2));   'If so call import function with new positions
    num$ := Print$("%d Features imported",n%);  'Update count of events
    ToolbarText(num$);      '..and display in toolbar
    o1Lev := HCursor(1);        'Get new positions
    o2Lev := HCursor(2);
    endif;
else
    if HCursor(1) <> o1Lev then     'Check for threshold movement
    n% := Import%(HCursor(1));      'Call import function with a new threshold position
    num$ := Print$("%d Features imported",n%);  'Update count of events
    ToolbarText(num$);          'Display in toolbar
    o1Lev := HCursor(1);        'Get new position
    endif;
endif;
return 1;
end;


Func Done%()
ToolbarClear(0);    'Clear the toolbar
Halt;
return 1;
end;

Func Import%(lev)
var peaks%;
View(data%);
MemDeleteItem(pChan%,-1);       'Clear existing items from the event channel
peaks% := MemImport(pChan%,wChan%,sTime,eTime,mode%,minInt,lev);        'Import items based on current settings
return peaks%;
end;

Func ChangeInt%()
var ok%;
DlgCreate("Minimum interval between events");
DlgReal(1,"Minimum interval",0,10);     'Re-set the minimum interval between events
ok% := DlgShow(minInt);
if mode% < 2 then       'If import mode requires an amplitude
    n% := Import%(HCursor(1)-HCursor(2));       'Use amplitude setting
    num$ := Print$("%d Features imported",n%);
    ToolbarText(num$);  'Update display count of events
else                    'If mode requires a threshold
    n% := Import%(HCursor(1));      'Use threshold setting
    num$ := Print$("%d Features imported",n%);
    ToolbarText(num$);      'Update display count of events
endif;
return 1;
end;


david

Scripts: Signal  

Q. I am using the MagStim support in Signal to control the various pulse parameters 
that I use during an experiment. I notice that the MagStim parameters used for 
each frame are stored as frame variables and that these can be accessed via the 
script language. Do you have an example script that can access and display 
these values? 

A. The attached script, FrameVariables.sgs, can read back MagStim or other 
auxiliary device parameters that are stored as frame variables in the Signal data 
file during sampling. This example prints a list of the variables for each frame to 
the Log window but could easily be modified to print specific variable values to an 
XY plot, for example. 

Did you know…? 
When using an output sequence you can disable interactive control of sequence jumps to 
prevent accidental changes caused by a user key press or mouse click in the sequencer 
control toolbar. These sequencer control options can be set from the drop-down list in the 
Sequencer tab of the sampling configuration. 

 
Sequencer jump control options 

Recent questions  
Q. I would like a way of setting an alarm clock facility in Spike2 to alert me when a 

set amount of time has elapsed during sampling.  

A. The attached script, Alarm.s2s, allows the user to set an alarm time based on 
the system clock of the computer. This example simply emits a tone and displays 
a message box once the alarm time is reached, but could easily be modified to 
perform other functions in response to the alarm setting such as starting 
sampling early in the morning before you arrive at work! 

User group     

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk.

 
All Trademarks are acknowledged to be the Trademarks of the registered holders.  

'$FrameVariables|Print frame variable values to the Log window from all frames in a data file

var dataFile%;		'Handle of data file
var numFrames%;		'Number of frames
var noVars%;		'No of variables for each frame
var vType%;			'Variable type
var varVal%;		'Variable values (integer, real or string)
var varVal;			
var varVal$;
var c%;				'Counter for frames
var v%;				'Counter for variables
var name$;			'Variable name
var units$;			'Units

dataFile% := FrontView();		'Show an open data file

if ViewKind(dataFile%) <> 0 then
	Message("    No Signal data file open\nPlease select a data file");
	dataFile%:= FileOpen("",0,1);	'if no data file found, prompt user to open one
	
	if dataFile% < 0 then				'If user cancels or error occurs
   		Message("	Unable to open a file! Re-run script to try again");		'Warn user 		
		Halt;		'and quit script	
	endif;
endif;
Window(0,0,100,100);	'Make the data full screen

numFrames%:= FrameCount();		'Get no. of frames
PrintLog("Frame\tVariable\tName\tValue\tUnits\n");	'Print headers to Log view

for c%:=1 to numFrames% do		'Cycle through frames
	Frame(c%);					'View current frame
	noVars%:= FrameVarCount();		'Get no. of frame variables

	for v%:=1 to noVars% do
		vType% := FrameVarInfo(v%,name$);	'Get name and info
		
		docase
			case vType% = 0 then		'If variable is an integer type
				varVal%:= FrameGetIntVar(name$,vType%,units$);		'Get values
				PrintLog("%d\t%d\t%s\t%d\t%s\n",c%,vType%,name$,varVal%,units$);	'Print to Log

			case vType% = 1 then		'If variable is a real type
				varVal:= FrameGetRealVar(name$,vType%,units$);		'Get values
				PrintLog("%d\t%d\t%s\t%f\t%s\n",c%,vType%,name$,varVal,units$);		'Print to Log

			case vType% = 2 then		'If variable is a String
				varVal$:= FrameGetStrVar$(name$,vType%,units$);		'Get values
				PrintLog("%d\t%d\t%s\t%s\t%s\n",c%,vType%,name$,varVal$,units$);	'Print to Log
		 endcase
	next;
PrintLog("\n");
next;

FrontView(LogHandle());		'Show Log view

'$Alarm clock| Example script that allows the user to set a time relative to the system
'clock for an alarm signal.

var ok%,dummy%,Hr%,Min%,Sec%;
var tim$;
Var s%,m%,h%;
Var lastt;
Var field%;
var TDSec%,TdMin%,TDHr%;
Var Alarm%;
var ct%:=0;
var today%;
Var tomorrow%;

Alarmset%();        'Main function

Func Alarmset%()
tim$:=Print$("Time is %s",Time$(1,1+2+4+8));    'Get's current time
TimeDate(sec%, min%, Hr%);


Lastt:=sec%+(Min%*60)+(Hr%*3600);

Sec%:=0;
DlgCreate("Alarm setup");  'Start new dialog
DlgAllow(511, update%);
field%:=DlgText(Tim$,0,4);
DlgInteger(1,"Hour",0,23,0,0,1);
DlgInteger(2,"Minute",0,59,0,0,1);
DlgInteger(3,"Second",0,59,0,0,1);
DlgButton(0,"Cancel");
DlgButton(1,"OK");
ok% := DlgShow(Hr%,min%,Sec%); 

If ok%=0 then
    Message("Alarm not set");       'If cancel is pressed
endif

TimeDate(TDSec%,TdMin%,TDHr%);

'Account for going over mignight
Today%:=(TDsec%+(TDMin%*60)+(TDHr%*3600));
Tomorrow%:=(sec%+(Min%*60)+(Hr%*3600));

Alarm%:=Tomorrow%-Today%;
If alarm%<0 then
    tomorrow%:=(86400-today%)+tomorrow%;
    Alarm%:=Tomorrow%;
endif

Seconds(0);
Return 1
end

DoToolbar();   'Try it out
Halt;

Func DoToolbar()    'Set your own name...
ToolbarClear();     'Remove any old buttons
ToolbarSet(0, "", AlarmIdle%);   'Idle routine
ToolbarSet(1, "&Quit");    'This button returns its number
return Toolbar("Waiting", 1023);
end;

Func AlarmIdle%()   'Button 0 routine
if ct%=0 then
    If seconds()>= Alarm% then      'Once alarm time is reached
        Sound(60,2,1);              'Play a tone
        Message("Turn off Alarm!"); 'And flash message
        ct%:=1;
    Endif
    
endif
return 1;  'This leaves toolbar active
end;

Func update%();
TimeDate(sec%, min%, Hr%);
If sec%+(Min%*60)+(Hr%*3600)> lastT then
    tim$:=Print$("Time is %s",Time$(1,1+2+4+8));
    Dlgvalue$(field%,Tim$);
    Lastt:=sec%+(Min%*60)+(Hr%*3600);
endif
return 1
end


david

